
Software
To use the 6551 as a polled device the software is fairly simple. First the 6551 must be reset and the baud rate, character length 
and parity must be set.

; set the 6551 register addresses
 
A_rxd           = $F000 ; ACIA receive data port
A_txd           = $F000 ; ACIA transmit data port
A_sts           = $F001 ; ACIA status port
A_res           = $F001 ; ACIA reset port
A_cmd           = $F002 ; ACIA command port
A_ctl           = $F003 ; ACIA control port
 
; initialise 6551 ACIA
 
    STA A_res       ; soft reset (value not important)
    LDA #$0B        ; set specific modes and functions
                ; no parity, no echo, no Tx interrupt
                ; no Rx interrupt, enable Tx/Rx
    STA A_cmd       ; save to command register
                ; all the following 8-N-1 with the baud rate
                ; generator selected. uncomment the line with
                ; the required baud rate.
;   LDA #$1A        ; 8-N-1, 2400 baud
;   LDA #$1C        ; 8-N-1, 4800 baud
    LDA #$1E        ; 8-N-1, 9600 baud
;   LDA #$1F        ; 8-N-1, 19200 baud
    STA A_ctl       ; set control register
To send a character all we need do is wait for the Tx buffer to empty then write the byte.

; wait for ACIA and Tx byte
 
    PHA         ; save A
LAB_WAIT_Tx
    LDA A_sts       ; get status byte
    AND #$10        ; mask transmit buffer status flag
    BEQ LAB_WAIT_Tx ; loop if tx buffer full
 
    PLA         ; restore A
    STA A_txd       ; save byte to ACIA data port
To receive a character we need to wait for the Rx buffer to fill then read the byte.

; wait for ACIA and Rx byte
 
LAB_WAIT_Rx
    LDA A_sts       ; get ACIA status
    AND #$08        ; mask rx buffer status flag
    BEQ LAB_WAIT_Rx ; loop if rx buffer empty
 
    LDA A_rxd       ; get byte from ACIA data port

